No more tedious times tables to recite; try binary multiplication in 6502 assembly language.

Multiplying by 1’s and 0’s

Multiplication. It's a subject that
gave most of us untold misery
in elementary school. But now, hav-
ing memorized all of those cursed
multiplication tables, we can confi-
dently multiply anything by any-
thing. Right?

Yes, right, as long as the anythings
are decimal numbers. Unfortunately,
if you want to write an assembly lan-
guage program to perform a multipli-
cation on your Apple, AIM 65, TRS-80
or other microcomputer, you're back
at square one, because (a) the num-
bers being multiplied are binary val-
ues and (b) none of the popular eight-
bit microprocessors has a multiply in-
struction.

That's the bad news. The good
news is that multiplication tech-
niques are covered in virtually every
tutorial and textbook on assembly
language programming. Regardless
of which book you pick up, you'll
learn how it's possible to multiply
two binary values, using the add and
shift instructions of the microproces-
sor. The discussions of multiplication
vary in quality—depending on the au-
thor's inclination and the software or
hardware orientation of the book—
but typically these books give only a
summary of the fundamental princi-
ples and an example or two (usually,
just an eight-bit by eight-bit unsigned
multiply). From there on, readers are
left to their own devices.

A cop-out? Maybe. But most au-
thors (and, apparently, their publish-
ers) feel that their book must cover so
many topics that they can't afford to

devote much page space to a “'sim-
ple” task like multiplication.

Admittedly, a brief once-over will
suit the needs of the casual reader/
programmer, but what about those
who want to multiply numbers that
are longer than eight bits, or those
who want to multiply signed (two's
complement) numbers?

This article is intended to serve
both kinds of readers: beginners who
want an overview of multiplication
and serious programmers who need a
more detailed treatment than the
popular literature provides. The pro-
grams are written in 6502 assembly
language, but the accompanying text
and flowcharts should make them
readily convertible for use with any
of the other eight-bit microproces-
sors.

Back to Fourth Grade

Before we discuss multiplying bi-
nary numbers, it may be instructive
to review the mechanics of multiply-
ing decimal numbers, the way we
used to do it in elementary school
with pencil and paper. As you will re-
call (in these days of calculators, it
may be a bit hazy), with pencil and
paper you write the multiplicand on
one line and the multiplier on the line
below it, and then grind out a series
of multiplications—one for each digit
in the multiplier. Each partial prod-
uct is recorded directly below its
multiplier digit, causing it to be offset
one digit position to the left of the
preceding partial product. When all
of the partial products have been cal-

culated, they are added to produce
the final product.

For example, multiplying the num-
ber 124 by the number 103 looks like
this:

124 Miltiplicand
x 103 Multiplier
7372 Partial Product #1
000 Partial Product #2

124 Partial Product #3

12772 Final Product
(Of course, you don't normally write
down the all-zeroes partial product,
but rather just skip to the next digit
position.)

It's important to remember why
we write the partial products offset
from each other: it's because each
partial product is associated with a
multiplier digit of a different decimal
weight. Remember that the preced-
ing problem can also be written in the
form
103% 124 =(3 x 124) + (0 x 124] + (100 x 124)
Or, we could use an equivalent form
103x124=(3x1x124)+(0x10x124)+(1 x
100 x 124)
to illustrate that the digits 3, 0 and 1
in the multiplier have decimal
weights of 1, 10 and 100, respective-
ly.

In the preceding problem, both the
multiplier and the multiplicand hap-
pen to be nonnegative (positive or un-
signed) numbers. How would the
pencil-and-paper operation have
changed if one or both was a negative

=
0000 MPLR=$20
0000 MPND=$21
0000 PROD=$22
0000 *=5400
0400 A9 00 MLT8U LDA #0
0402 A2 08 LDX 48
0404 46 20 NXTBT LSR MPLR
0406 90 03 BCC ALIGN
0408 18 CLC
0409 65 21 ADC MPND
040B 6A ALIGN ROR A
040C 66 22 ROR PROD
040E CA DEX
040F DO F3 BNE NXTBT
0411 B5 23 STA PROD+1
0413 60 RTS

This subroutine multiplies an 8-bit unsigned multiplicand (MPND)
by an B8-bit unsigned multiplier (MPLR), and returns the 16-bit
product in locations PROD (low byte) and PROD+1 (high byte).

Example 1. An eight-bit by eight-bit unsigned multiplication subroutine.

Clear product MSBY
Multiplier bit count = 8
Get next multiplier bit
Multiplier bit = 1?2

Yes, add multiplicand
to partial product
Rotate product right

Loop until 8 bits are done
Store product MSBY

number? Very little, because we all
know that if one number is positive
and the other is negative, the product
will be negative—so you tack a minus
sign onto the answer. Similarly, if
both multiplier and multiplicand are
negative, the product will be positive
—s0 you omit the minus sign from
the answer (or put a plus sign on it, if
you're a purist]. With binary num-
bers, though, there's quite a bit of dif-
ference between multiplying un-
signed numbers and multiplying
signed numbers, because signed
numbers are represented in two's
complement form. We'll be looking
at both unsigned and signed multipli-
cation later, but for now, let's briefly
discuss how binary numbers can be
multiplied.

Binary Multiplicaton vs.
Decimal Multiplication

Binary multiplication is much sim-
pler than decimal multiplication, be-
cause binary multipliers consist of
only the digits 0 and 1, whereas deci-
mal multipliers can be made up of the
digits 0 through 9. In binary multipli-
cation, the partial product will al-
ways be simply a copy of the multi-
plicand if the multiplier digit is 1, and
it will be 0 if the multiplier digit is 0.

The binary equivalent of our previ-
ous 103 x 124 example looks like this:

01111100 Multiplicand (= 124)
x 01100111 Multiplier (=103)
01111100
01111100
01111100
00000000
00000000
01111100
01111100
00000000
011000111100100  Final Product (=12772)
This example gives a good indica-

tion of the way in which eight-bit mi-
croprocessors must perform multipli-
cation operations. There are some
important differences, however.
When performing a binary multipli-
cation by hand, you must calculate
the final product by adding the par-
tial products, column-by-column.
(And if you think that's easy, try it!
The carries can drive you bonkers.)
The operation is similar on a comput-
er, but instead of waiting until all of
the individual partial products are
calculated before deriving the final
product, computer programs update
the partial product after each multiplier
bit is examined. By doing this, the final
product is generated when the last bit
of the multiplier has been processed.
Moreover, when multiplying by
hand, each partial product is offset
one digit position to the left of the
preceding partial product, to account
for the weight of the multiplier bit. In
a computer, it is easier to shift the
partial product each time it is updat-
ed, thereby aligning it to receive the
contribution of the next multiplier
bit. The partial product may be shift-
ed either right or left, depending on
whether your program is examining
the multiplier bits from right to left
(low-order to high-order) or from left
to right (high-order to low-order). In
this article, the multiplier will be pro-
cessed right-to-left, the way you
would do it by hand, so the partial
product will be shifted to the right.
In summary, the following applies
when multiplying binary numbers
by a computer:
If the multiplier bit is a 1, add the multi-
plicand to the partial product, and then
shift the sum one bit position to the
right. If the multiplier bit is a 0, shift the
current partial product one bit position

to the right, with no addition.

In writing a multiplication program
for your microcomputer, which in-
struction would you expect to use to
perform the add and right-shift oper-
ations? With a 6502-based microcom-
puter, such as Apple II, KIM-1,
SYM-1 or AIM 65, the addition will
be performed with the 6502's only
add instruction, add to accumulator
with carry (ADC). The shifting will
be performed with the shift right
(LSR) or rotate right (ROR| instruc-
tion.

With this background, let's exam-
ine the software that will be needed
to multiply unsigned or signed num-
bers.

Multiplying Unsigned Numbers

As you probably know, in an un-
signed number, each data bit carries
a certain binary weight, according to
its position within the number. Data
bits are numbered from right to left,
with the rightmost bit labeled as bit 0
and the leftmost bit labeled bit 7. The
bit numbering scheme has a direct
correlation to the binary weights, in
that bit 0 has a weight of 2° (decimal
1) and bit 7 has a weight of 27 (deci-
mal 128). Therefore, a single byte can
represent an unsigned number from
decimal 0 (binary 00000000) to deci-
mal 255 (binary 11111111).

Single-Precision
Unsigned Multiplication

Certainly, the easiest place to begin
is by developing a program—a sub-
routine, actually—to multiply two
eight-bit unsigned numbers in mem-
ory. If the multiplicand and the mul-
tiplier are both eight bits long, how
long will the product be? Well, we
know that the worst case involves
multiplying 255 by 255, which gives
a product of 65,025. To hold a value
of 65,025, we need 16 bits, or two
bytes.

At this point, we can draw the flow-
chart that will serve as the blueprint
for an eight-bit (or single-precision)
multiplication subroutine. This flow-
chart must do five things:

1. Initialize a two-byte product in
memory to zero, and a multiplier bit
counter to eight.

2. Shift the multiplier right one bit
position into carry.

3. Interrogate the state of the multi-
plier bit that's in carry. If carry=1,
add thc multiplicand to thc partial
product.

4. Rotate the partial product right,
into the final product location's least-

PROD+2 and PROD+3 (high byte).
0000 MPLR=$20
0000 MPND=$22
0000 PROD=$24
0000 *=5400
0400 A9 00 MLT16 LDA #0
0402 85 26 STA PROD+2
0404 85 27 STA PROD+3
0406 A2 10 LDX #16
0408 46 21 NXTBT LSR MPLR+1
040A 66 20 ROR MPLR
040C 90 OB BCC ALIGN
040E A5 26 LDA PROD+2
0410 18 CLC

0411 65 22 ADC MPND
0413 85 26 STA PROD+2
0415 A5 27 LDA PROD+3
0417 65 23 ADC MPND+1
0419 6A ALIGN ROR A
041A 85 27 STA PROD+3
041C 66 26 ROR PROD+2
041E 66 25 ROR PROD+1
0420 66 24 ROR PROD
0422 CA DEX

0423 DO E3 BNE NXTBT
0425 60 RTS

This subroutine multiplies a 16-bit unsigned multiplicand (MPND,
MPND+1) by a 16-bit unsigned multiplier (MPLR, MPLR+1), and
returns the 32-bit product in locations PROD (low byte), PROD+1,

Clear P2 and P3 of product

Multiplier bit count = 16
Get next multiplier bit
into Carry

Multiplier bit = 12

Yes, fetch P2

and add MO0 to it

Store new P2

Fetch P3

and add Ml to it
Rotate product right

Decrement bit count
Loop until 16 bits are done

Exampie 2. A 16-bit by 16-bit unsigned multiplication subroutine.

significant byte (LSBY).

5. Decrement the multiplier count.
If it is zero, store the final product's
most-significant byte (MSBY) in
memory, and return; otherwise, go
back to process the next multiplier
bit.

The flowchart in Fig. 1 performs
these five tasks.

PRODUCT » @
COUNT - &

SHIFT MULTIPLIER
RIGHT, INTO CARRY

CLEAR CARRY

ADD MULTIPLICAND
TO PRODUCT MSBY

ROTATE PRODUCT
RIGHT

COUNT = COUNT = 1

Es
sev

STORE PRODUCT
»

Fig. 1. An eight-bit by eight-bit unsigned multipli-
cation algorithm.

Example 1 is a subroutine (MLT8U)
that uses the flowcharted algorithm
to multiply the contents of a multipli-
cand in memory (MPND, assigned to
location $21 here) by the contents of a
multiplief in memory (MPLR, as-
signed to location $20 here|. The
16-bit product is returned in two con-
secutive locations, PROD and
PROD +1. The X register is used to
hold the multiplier bit count.

In the MLT8U subroutine, the LSR
MPLR instruction at NXTBT causes
the multiplier (in memory location
$20) to be shifted, one bit at a time,
into carry. If the shifted multiplier bit
isa one, the CLC and ADC MPND in-
structions add the multiplicand to the
most-significant byte of the product,
and the two rotate instructions at
ALIGN (ROR A and ROR PROD) shift
the partial product to the right, into
the least-significant byte. If the shift-
ed bit of the multiplier is a zero, BCC
ALIGN bypasses the add-multipli-
cand operation by branching to the
rotate-right sequence at ALIGN. The
NXTBT loop is executed eight times,
once for each bit in the multiplier.

Double-Precision
Unsigned Multiplication

We've just concluded a discussion
of eight-bit by eight-bit unsigned
multiplication, which lets us multiply
two numbers as large as 255. Unfor-
tunately, like a three-legged race
team in the Boston Marathon, single-
precision multiplication is nice, but

Pz, P30
COUNT « 18
SHIFT MULTIPLIER
RIGHT, INTO CARRT

ROTATE P2, P1, PO
RIGHT

COUNT + COUNT- |

Fig. 2. A 16-bit by 16-bit unsigned multiplication
algorithm

not very practical, since many appli-
cations require numbers larger than
255 to be multiplied. Obviously, the
next step is to develop a program that
will multiply 16-bit, or double-preci-
sion, numbers.

Multiplying 16-bit numbers is
somewhat more complex than multi-
plying eight-bit numbers, due to the
additional memory involved, but the
add-and-shift procedure still applies.
With a double-precision multiplica-
tion, the multipler and multiplicand
are both 16-bit values, so the product
will occupy 32 bits (or four bytes) in
memory.

Fig. 2 is a flowchart for a double-
precision unsigned multiplication
subroutine. The two-byte multipli-
cand is represented by the symbols
MO (low order byte) and M1 (high-or-
der byte). The four-byte product is
represented by the symbols PO (low-
order byte), P1, P2 and P3 (high-order
byte). The double-precision algo-
rithm shown in Fig. 2 operates simi-
larly to the single-precision algorithm
that has just been discussed. That is,
the multiplicand is added to the high-
order half of the partial product (P2
and P3, here) if carry is a 1. The result
is then rotated one bit to the right,

Fig. 3. Multiplier, multiplicand and product in memory.

along with the low half of the partial
product.

Example 2 is a subroutine (MLT16)
that uses the flowcharted algorithm
(Fig. 2) to perform a double-precision
unsigned multiplication. In this par-
ticular subroutine, the multiplier
(MPLR) is in locations $20 and $21,
the multiplicand (MPND) is in loca-
tions $22 and $23 and the 32-bit prod-
uct is returned in locations $24 (low
byte) through $27 (high byte). Each
time a multiplier bit is a one, the
MLT16 subroutine must perform two
adds, to add the multiplicand to the
partial product, and two stores, to up-
date the product in memory. It must
also perform four rotate operations
(one for each byte in the product).
Note that the updated byte P3 is re-
turned to memory after the first of the
four rotates.

Whither Goest the Multiplier?

If you've run either Example 1 or
Example 2 on a microcomputer,
you've observed that the multiplier
has been cleared to zero in the course
of the operation. In many applica-
tions, it makes no difference whether
or not the multiplier is affected. In
other applications, the multiplier
must be preserved for one reason or
another.

How can the multiplier be pre-
served? Typically, the first impulse is
to preserve it by simply changing the
LSR instruction at ALIGN to an ROR
instruction. Well, that's a good begin-
ning, but you must be careful to ob-
serve that for a rotate to work correct-
ly, the sense of the carry bit must be
unchanged from one rotate operation
to another. Unfortunately, the add
operation between labels NXTBT
and ALIGN will always alter the carry
bit!

With these two considerations in
mind, we're on our way to a solution.
So far, we've discovered that if you

want to preserve the multiplier you
must

1. Rotate, rather than shift, the
multiplier, and

2. Save the state of the carry bit be-

tween rotate operations.
Will these two meodifications alone
preserve the multiplier? Well, al-
most. They don't quite do the job be-
cause the final rotate operation will
leave the most-significant multiplier
bit in carry, and all other less-signifi-
cant multiplier bits displaced one bit
position to the left. So, in addition,
you must

3. Rotate the multiplier one more
time, at the end of the operation.

By applying these three rules to Ex-
ample 2, you may come up with a
subroutine that looks like the one in
Example 3, a double-precision un-
signed multiplication subroutine in

which the multiplier is preserved.
Since the multiplier must be rotated
at two different points in the program
(per rules 1 and 3}, the rotate instruc-
tions are given in a subroutine, called
RMPLR. Rule 2 is easily satisfied by
executing a push processor status
(PHP) instruction after the call to
RMPLR, and a complementary pull
processor status (PLP) instruction just
after the partial product is rotated.

Multiprecision Unsigned
Multiplication

Many real-world applications in-
volve multiplying numbers that are
longer than two bytes, or have a mul-
tiplier and multiplicand of different
lengths. For these reasons, it is worth-
while to wind up this discussion of
unsigned multiplication by develop-
ing a subroutine that can multiply un-
signed numbers of any length. That
is, we will develop a subroutine that
multiplies an M-byte multiplicand by
an N-byte multiplier, and yields an
|N +M)-byte product. Fig. 3 shows
how these terms are stored in mem-

The overall approach is unchanged
for this general case, but since the
multiplier and multiplicand are vari-
able-length values, we will have to
compute parameters that were known
in the eight-bit and 16-bit multiplica-
tions. Here, for example, the number

This subroutine is a modified version of the subroutine in
Example 2. It has some additional instructions to return the
multiplier in its original form.
0000 MPLR=$20
0000 MPND=$22
0000 PROD=$24
0000 *=5400
0400 A9 00 MLT16 LDA #0 Clear P2 and P3 of product
0402 B85 26 STA PROD+2
0404 85 27 STA PROD+3
0406 A2 10 LDX #16 Multiplier bit count = 16
0408 18 CLC
0409 20 27 04 NXTBT JSR RMPLR Go get next multiplier bit
040C 08 PHP and save it (carry)
040D 90 OB BCC ALIGN Multiplier bit = 1?2
040F A5 26 LDA PROD+2 Yes, fetch P2
0411 18 CLC and add M0 to it
0412 65 22 ADC MPND
0414 85 26 STA PROD+2 Store new P2
0416 A5 27 LDA PROD+3 Fetch P3
0418 65 23 ADC MPND+1 and add M1 to it
0412 6A ALIGN ROR A Rotate product right
041B 85 27 STA PROD+3
041D 66 26 ROR PROD+2
041F 66 25 ROR PROD+1
0421 66 24 ROR PROD
0423 28 PLP Retrieve Carry from stack
0424 ca DEX Decrement bit count
0425 DO E2 BNE NXTBT Loop until 16 bits are done
0427 66 21 RMPLR ROR MPLR+1 Rotate multiplier right
0429 66 20 ROR MPLR
042B 60 RTS
Example 3. A double-precision unsigned multiplication subroutine that preserves the multiplier.




of bits in the multiplier must be de-
termined by multiplying N by eight
(with three left-shifts). The length of
the product must also be computed,
by adding N and M. And whenever
the multiplicand needs to be added to
the partial product, it will have to be
added to the most-significant '"M"
bytes—still another required compu-
tation. These various computations
are reflected in Fig. 4, the flowchart
for multiprecision unsigned multipli-
cation.

As you can see, the flowchart in
Fig. 4 contains the same number of
steps as the flowchart for the double-
precision case (Fig. 2). That's not sur-
prising, since the same types of oper-
ations are being performed, but how
do the actual multiplication subrou-
tines compare? That is, what pro-
gramming overhead is involved in
having a general-purpose subroutine
rather than a length-specific subrou-
tine? The answer is apparent by com-
paring Example 4, the multiprecision

This subroutine multiplies two variable-length, unsigned
integers. The multiplier is stored starting at location
MPLR, and is N bytes long. The multiplicand is stored
starting at location MPND, and is M bytes long. The
product will be "N + M" bytes long, and will be returned
in memory starting at location PROD.
This subroutine affects the A, X and Y registers.
0000 N=$20 Multiplier length
0000 M=$21 Multiplicand length
0000 MPLR=$30 Multiplier loc.
0000 MPND=$40 Multiplicand loc.
0000 PROD=$50 Product loc.
0000 PINDX *=*+1 Product index
0001 MBIT *=#*+1 Multiplier bit count
0002 *=$600
0600 18 MMPYU CLC Calculate product index
0601 A5 20 LDA N (N+M-1)
0603 AR TAX
0604 65 21 ADC M
0606 A8 TAY and save it in Y
0607 88 DEY
0608 84 00 STY PINDX and in PINDX
060A A9 00 LDA #0 Clear the high M bytes
060C 99 50 00 CLRP STA PROD,Y of the product
060F 88 DEY
0610 C4 20 CPY N
0612 BO F8 BCS CLRP
0614 8A TXA Calculate multiplier bit count
0615 0A ASL A
0616 OA ASL A
0617 OA ASL A
0618 85 01 STA MBIT and save it in MBIT
061A 20 4A 06 NXTBT JSR RMPLR Get next multiplier bit
061D 08 PHP Save resulting Carry
061E 90 El1 BCC ALIGN Multiplier bit = 12
0620 A4 20 LDY N Yes. Add multiplicand to
0622 A2 00 LDX #0 high M bytes of product
0624 18 CcLC
0625 B9 50 00 AMPND LDA PROD,Y
0628 75 40 ADC MPND,X
062A 99 50 00 STA PROD,Y
062D C8 INY
062E E8 INX
062F 08 PHP Save Carry between adds
0630 C4 00 CPY PINDX
0632 FO 04 BEQ PULLC
0634 90 02 BCC PULLC
0636 BO 04 BCS GOROT
0638 28 PULLC PLP
0639 4C 25 06 JMP AMPND
063C 28 GOROT PLP
063D 18 CLC
063E A6 00 ALIGN LDX PINDX
0640 76 50 RPROD ROR PROD,X
0642 CA DEX
0643 10 FB BPL RPROD
0645 28 PLP Retrieve Carry from stack
0646 C6 01 DEC MBIT Multiplier fully processed?
0648 DO DO BNE NXTBT No. Loop for next bit
064A A6 20 RMPLR LDX N Yes. Rotate multiplier right
064C CA DEX
064D 76 30 RBYTE ROR MPLR,X
064F CA DEX
0650 10 FB BPL RBYTE
0652 60 RTS

Example 4. A multiprecision unsigned multiplication subroutine.

unsigned multiplication subroutine,
with Example 3, the 16-bit multipli-
cation subroutine. The multipreci-
sion subroutine is twice as long as the
double-precision subroutine!

Although Example 4 looks compli-
cated, it isn't. Its initialization in-
cludes two additional input parame-
ters {the multiplier length, N, and the
multiplicand length, M) and two
symbolic locations (to hold the prod-
uct index, PINDX, and the multiplier
bit count, MBIT). The subroutine it-
self differs very little from Example
3, except for the various computa-
tions and the use of indexes for ad-
dressing.

What lengths of numbers can be
multiplied by the subroutine shown
in Example 47 Well, you can see that
the first seven instructions add N and
M, and put the result in an eight-bit
register (Y) and an eight-bit memory
location (PINDX); therefore, the sum
of N and M must not exceed decimal
255. The multiplier bit count (8 x N} is
also stored in an eight-bit memory lo-
cation, so the value of N must not ex-
ceed 31. These two limitations allow
us to conclude that the multiplicand
length, M, must not exceed (255-31)

correlation to the binary weights in that bit 0
has a weight of 2¢ (decimal 1), bit 1 has a
weight of 2! (decimal 2}, and so on. Thus, bit 7
has a weight of 27 (decimal 128). The assign-
ments can be summarized as follows:

7 65 4321 0 Bitposition

2T 20 2% 2422 22 2 2° Binary weight

128 64 32 16 8 4 2 | Equivalent decimal weight
As you can see, a single byte can representan
unsigned number from 0 (binary 00000000|
to decimal 255 (binary 11111111).

In a signed number, the seven low-order
bits (bits 0 through 6) represent data, and
have the same relative weights as the bits in
unsigned numbers. The most-significant bit
(bit 7) represents the sign of the number. 1f
the number is positive or zero, bit 7 is a logic
0. If the number is negative, bit 7 is a logic 1.
A single byte can represent a positive signed
number from 0 (binary 00000000) to +127
(binary 01111111}, or a negative signed num-

binary combination 00000000 (the positive
form) and binary combination 10000000 (the
negative form). Using two's complement,
zero is represented by only one form, the bi-
nary combination 00000000.

To derive the negative two's complement
form of a binary number, you simply take the
positive form of the number, reverse the
sense of each bit (change each 1 to a 0, and
each 0to a 1) and add 1 to the result. The fol-
lowing example shows the steps required in
deriving the binary representation of —32 in
two’s complement form:
+00100000 +32

11011111 One's complement

+ 1 Add 1
11100000 -32 in two's complement form

From the book 6502 Software Design by Leo J.
Scanlon. It is reproduced here with permussion of
the publisher, Howard W. Sams & Co., Inc.

